# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=perl-path-isdev
pkgver=1.001003
pkgrel=6
pkgdesc="Determine if a given Path resembles a development source tree"
arch=('any')
license=('PerlArtistic')
url="https://metacpan.org/release/Path-IsDev"
depends=('perl-class-tiny' 'perl-file-homedir' 'perl-module-runtime' 'perl-path-tiny'
         'perl-role-tiny' 'perl-sub-exporter')
makedepends=('perl-test-fatal')
options=('!emptydirs')
source=("https://cpan.metacpan.org/authors/id/K/KE/KENTNL/Path-IsDev-$pkgver.tar.gz")
sha512sums=('fd4e355d0e67ec210eba443d5867dce181664da6ae300d81b98e77fd78a25d5a4f95e0b901f0a6c7596f28a3e2e37264f31226c410f5e6fb6a565d2ef0618d59')

build() {
  cd Path-IsDev-$pkgver
  perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd Path-IsDev-$pkgver
  make test
}

package() {
  cd Path-IsDev-$pkgver
  make DESTDIR="$pkgdir" install
}
